

```

1 testcase validWithdrawal (inout reason_Type reason) runs on hardwareEmulator_CType system
2 ATM_Interface_CType {
3
4     var boolean result;
5     var operationComplete_Type erroneousWithdrawal;
6     var default theDefault := activate(HWE_Default());
7
8     // Creation of PTC
9     var bankEmulator_CType BE_PTC:=bankEmulator_CType.create;
10
11    // Mapping & Connecting ports
12    map(system:niSUT, BE_PTC:niCom);
13    connect(self:coHWE, BE_PTC:coBE);
14    map(system:hwiSUT, self:hwiCom);
15
16    BE_PTC.start(BE_validWithdrawal()); // start PTC
17    testCaseGuard.start; // start guarding timer
18
19    // invocation of authentication procedure
20    result := authentication(validCard_Par, validPin_Par, reason);
21
22    if (result != true) { // test case fails in authentication procedure
23        setverdict(inconc); BE_PTC.stop; // stop the bank emulator test component
24        stop;
25    }
26
27    // start of withdrawal procedure
28    hwiCom.send(validWithdrawalOp_Template);
29    reason := unknown; // return value if MTC fails in default
30    alt {
31        [] hwiCom.receive(operationComplete_Type:{withdrawal, true, ?})
32        {
33            reason := noReason;
34            setverdict(pass);
35            all component.done; // wait for the termination of PTC
36        }
37        [] hwiCom.receive(operationComplete_Type:?) -> value erroneousWithdrawal {
38            setverdict(fail);
39            reason := erroneousWithdrawal.reason;
40        }
41        [] coHWE.receive(reason_Type:?) -> value reason {
42            setverdict(fail);
43        }
44        [] testCaseGuard.timeout {
45            reason := notInTime; setverdict(fail);
46        }
47    }
48    deactivate(theDefault);
49    if (reason != noReason) { // Stopping PTC in case of a failure
50        BE_PTC.stop;
51        stop;
52    }
53 }

```